

Robot Arm

Design Brief

Design a simple robotic arm that can follow a sequence of precise movements.

Circuit Explanation


The circuit controls three radio-control style servos via outputs 0 to 2. Note that servos generate a large amount of electrical noise and so it is essential they are driven by a separate 6V power supply.

Three potentiometers are connected to the analogue inputs of the PICAXE so the servos can be manually controlled for testing purposes. Note that a serial LCD connected here could help testing here – the value from the analogue input could be shown on the serial LCD so that the correct value for programming is then known.

Program Explanation

The first program reads the analogue values from the potentiometers and transfers them to the servo commands to move the servos. Note that a time delay is included to decrease the refresh rate of the servo commands, which could cause the servos to 'jitter' slightly.

The second program shows how a sequence of steps can be programmed by use of the servo commands. Note that the servo values should only ever be in the range 75 to 255, values outside this range could possibly damage the servos.



Program Listing

```
` Robot Arm using servos
` For PICAXE-28

` Three presets on analogue 0-2
` Three servos on output 0-3

` This program reads the three presets and moves
` the servo according to the preset value.
` As servos only react to pulses between 75-225
` a check is made to adjust value to these boundaries.

symbol pos0 = b1      ` servo 1 position
symbol pos1 = b2      ` servo 2 position
symbol pos2 = b3      ` servo 3 position

` *** read the three analogue inputs ***

main:
    readadc 0, pos0
    readadc 1, pos1
    readadc 2, pos2

` *** test all 3 values are between 0 and 225 ***
` *** if not correct to exactly 75 or 225 ***
test0:
    if pos0 < 75 then pos0low
    if pos0 > 225 then pos0high
test1:
    if pos1 < 75 then pos1low
    if pos1 > 225 then pos1high
test2:
    if pos2 < 75 then pos2low
    if pos2 > 225 then pos2high

pulses:
`*** optionally display values on serial LCD
`    serout 7, N2400,(254,128,"0=",#pos0," ")
`    serout 7, N2400,(254,136,"1=",#pos1," ")
`    serout 7, N2400,(254,192,"2=",#pos2," ")
`***

` *** now set servo pulses ***
    servo 0, pos0
    servo 1, pos1
    servo 2, pos2

    pause 1000
    goto main

` *** corrections. Note how each test ***
` *** jumps back to next test above ***

pos0low:
    let pos0 = 75
    goto test1
pos0high:
    let pos0 = 225
    goto test1
pos1low:
    let pos1 = 75
    goto test2
pos1high:
```

```
        let pos1 = 225
        goto test2
pos2low:
    let pos2 = 75
    goto pulses
pos2high:
    let pos2 = 225
    goto pulses

` Robot Arm using servos
` For PICAXE-28

` Three presets on analogue 0-2
` Three servos on output 0-3

` *** reset all servos upon a power reset ***

reset:
    servo 0, 150
    servo 1, 150
    servo 2, 150

    pause 2000

` *** loop of robot arm movements ***

loop: servo 0, 180
    pause 1000
    servo 1, 75
    servo 2, 100
    pause 2000
    servo 0, 150
    pause 1000
    servo 2, 225
    pause 3000
    servo 1, 150
    servo 2, 150
    pause 2000
    goto loop
```