
Snooker Game
Design Brief
Design an electronic snooker scoring system.

Circuit Explanation
A series of coloured switches are used to represent each colour
ball. A slide switch is used to switch between player 1 and player
2, and an LED indicates which player is selected. A serial LCD
module is used to display the snooker score.

Program Explanation
The scores for the two players are scored within two variables b0
and b1. When a switch is pushed the correct value (1 to 7) is
added to the appropriate players score. This score is then updated
and then shown on the serial LCD module.

Program Listing
‘ Snooker Scoring Device
‘ For PICAXE-28

init: let b1 = 0 ‘ reset player 1 score
let b2 = 0 ‘ reset player 2 score
serout 7,N2400,(254,1) ‘ clear LCD display
pause 30 ‘ short delay for LCD

main: ‘ update scores on LCD
serout 7,N2400,(254,128,”Player 1 = “,#b1,”  “)
serout 7,N2400,(254,192,”Player 2 = “,#b2,”  “)

let b3 = 0 ‘ reset new ball value to 0

‘ now loop checking all the colour switches
loop: if pin7 = 1 then black

if pin6 = 1 then pink
if pin5 = 1 then blue
if pin4 = 1 then brown
if pin3 = 1 then green
if pin2 = 1 then yellow
if pin1 = 1 then red

‘ also check player switch for correct LED

if pin0 = 1 then light2 ‘ jump if player 2

light1: high 1 ‘ light player 1 LED
low 2 ‘ player 2 off
goto loop ‘ keep looping

light2: high 2 ‘ light player 2 LED
low 1 ‘ player 1 off
goto loop ‘ keep looping

‘ this section adds new ball score to variable b3
‘ note the multiple entry points
‘ to give correct value 1 to 7
‘ e.g. black(7) = 1+1+1+1+1+1+1

black: let b3 = b3 + 1 ‘ black = 7
pink: let b3 = b3 + 1 ‘ pink = 6
blue: let b3 = b3 + 1 ‘ blue = 5
brown: let b3 = b3 + 1 ‘ brown = 4
green: let b3 = b3 + 1 ‘ green = 3
yellow: let b3 = b3 + 1 ‘ yellow = 2
red: let b3 = b3 + 1 ‘ red = 1

sound 6,(50,100) ‘ make a beep sound

‘ check to see if added to player 1 or 2
if pin0 = 1 then addto2 ‘ jump if player 2

addto1: let b1 = b1 + b3 ‘ add score to player 1
goto main ‘ loop back & update LCD

addto2: let b2 = b2 + b3 ‘ add score to player 2
goto main ‘ loop back & update LCD


